3 1176 00148 9716

NASA TM- 81798

# NASA Technical Memorandum 81798

NASA-TM-81798 19800015510

# AN INTERACTIVE COMPUTER PROGRAM FOR SIZING SPACECRAFT MOMENTUM STORAGE DEVICES



F. J. Wilcox, Jr.



March 1980

National Aeronautics and
Space Administration

**Langley Research Center**
Hampton, Virginia 23665

# AN INTERACTIVE COMPUTER PROGRAM FOR SIZING SPACECRAFT MOMENTUM STORAGE DEVICES

Floyd J. Wilcox, Jr.
Langley Research Center

## SUMMARY

An interactive computer program has been developed which computes the sizing requirements for nongimbled reaction wheels, control moment gyros (CMG), and dual momentum control devices (DMCD) used in Earth-orbiting spacecraft. The program accepts as inputs the spacecraft's environmental disturbance torques, rotational inertias, maneuver rates, and orbital data. From these inputs, wheel weights are calculated for a range of radii and rotational speeds. The shape of the momentum wheel may be chosen to be either a hoop, solid cylinder, or annular cylinder. The program provides graphic output illustrating the trade-off potential between the weight, radius, and wheel speed. A number of the intermediate calculations such as the X-, Y-, and Z-axis total momentum, the momentum absorption requirements for reaction wheels, CMG's, DMCD's, and basic orbit analysis information are also provided as program output.

## INTRODUCTION

Earth-orbiting spacecraft utilize nongimbled reaction wheels, control moment gyros (CMG), and dual momentum control devices (DMCD) for momentum storage and control, and the development of accurate momentum wheel sizing requirements is essential for effective spacecraft design. As part of the Langley Research Center's Computer-Aided Spacecraft Design effort, an interactive computer program has been developed to size momentum wheels. The program accepts as input the spacecraft's environmental disturbance torques, rotational inertias, maneuver rates, vehicular orientation, and orbital data. Momentum wheel weights are calculated for a range of wheel radii and rotational speeds, and are provided as graphical output to illustrate the trade-off potential between the weight, radius, and wheel speed. Intermediate calculations such as the X-, Y-, and Z-axis total momentum, the momentum absorption requirements for reaction wheels, CMG's, DMCD's, and basic orbit analysis information are also provided as output.

The momentum wheel sizing computer program listing is given in the appendix.

## PROGRAM DESCRIPTION

The momentum wheel sizing computer program provides an interactive graphics technique for determining the wheel size and weight for various momentum storage

N80-24102 #

and control devices on Earth-orbiting spacecraft. The program is written in ANSI standard FORTRAN on a Control Data Corporation (CDC) CYBER 173 computer. The graphics output is generated using Tektronix Plot 10 software routines in conjunction with a Tektronix 4014/15 graphic terminal. The program should be easily adapted to and can be executed on any host computer with this graphics package and remote terminal.

A simplified flow diagram of the momentum wheel sizing program is shown in figure 1.

## Inputs and Assumptions

The user provides the following inputs: spacecraft environmental disturbance torques, rotational inertias, maneuver rates, orbit altitudes and inclination, vehicle orientation, and celestial orientation. These inputs are provided as outputs from two other Langley Research Center Computer-Aided Spacecraft Design programs, (1) the Spacecraft Design and Cost Model (SDCM), reference 1, and (2) the Large Area Space System (LASS), reference 2. The orbital equations of motion were obtained from reference 3. The relationship between the orbital reference frames and the vehicle reference frames is shown in figure 2. The input parameters are:

INOSE - Vehicle Pointing Orientation

1. Sideways
2. Nose Down
3. Nose Forward

ISATOR - Celestial Orientation

1. Earth
2. Solar
3. Inertial

XNNN - Number of Gimbled Configurations for CMG

A - Apoapsis Altitude

P - Periapsis Altitude

ORBINC - Orbit Inclination

TL - Number of Orbits Between Wheel Unloading

TACCEL - Maneuver Acceleration Time

PDOTR $\begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix}$ - X-, Y-, Z-Axis Spacecraft Maneuver Rate

UPSILN - Angular Pivoting for DMCD

TA $\begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix}$ - X-, Y-, Z-Axis Atmospheric Disturbance Torque

TG $\begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix}$ - X-, Y-, Z-Axis Gravity Gradient Disturbance Torque

TS $\begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix}$ - X-, Y-, Z-Axis Solar Disturbance Torque

$\begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix}$ J - X-, Y-, Z-Axis Spacecraft Rotational Inertia

A major assumption of the program is that the X-axis is the spin axis while the Y- and Z-axes are transverse. Also, the spacecraft's orbit may be either circular or elliptical. Although all the momentum control wheels are normally the same size, the X-, Y-, and Z-axes momentum absorption requirements are not identical, therefore, it is necessary to determine the wheel weight for the axis with the largest momentum absorption requirement.

## Orbit Analysis

The first step of the orbit analysis portion of the program is to analyze the ellipse representing the orbit, as shown in figure 3.

From the figure, the semimajor axis, a, is calculated by

$$a = \frac{(2R + A + P)}{2}$$

The radius at periapsis, RP, is calculated by

$$RP = R + P$$

the focus of the ellipse, c, is calculated by

$$c = a - RP$$

and the eccentricity, e, is calculated by

$$e = \frac{c}{a}$$

where R is the radius of the Earth.

In the case of a circular orbit, the semimajor axis becomes the circle's radius which is also the radius at periapsis.

$$a = RP$$

This forces both the focus and eccentricity to zero

$$c = 0$$

$$e = 0$$

The orbital period, T, is then calculated using Kepler's 3/2 Law

$$T = \frac{2a^{3/2}}{\sqrt{u}}$$

where u is Earth's gravitational constant.

The mean orbital motion, N, is calculated using

$$N = \sqrt{\frac{u}{a^3}}$$

and the eccentric anomaly, E, by

$$E = N + e \sin (N) + \frac{1}{2} e^2 \sin (2N)$$

In the case of a circular orbit, the last two terms of the preceding equation are forced to zero and the eccentric anomaly becomes

$$E = N$$

Using the above calculations, the maximum orbital angular rate of change, $\dot{\theta}$, which is at perigee, is determined from

$$\dot{\theta} = \frac{N\sqrt{1 - e^2}}{\left[1 - e \cos (E)\right]^2}$$

For a circular orbit, the maximum rate of change is the mean orbital motion such that

$$\dot{\theta} = N$$

## Total Momentum

To calculate the total momentum in each of the three spacecraft's body axes, three different momentums are taken into account: (1) maneuver momentum, (2) environmental disturbance torque momentum, and (3) orientation tracking momentum.

The maneuver momentums, HXMAN, HYMAN, and HZMAN, in all three axes are calculated using the following equations:

$$\text{HXMAN} = \frac{\text{XJ(PDOTRX)}}{57.3}$$

$$\text{HYMAN} = \frac{\text{YJ(PDOTRY)}}{57.3}$$

$$\text{HZMAN} = \frac{\text{ZJ(PDOTRZ)}}{57.3}$$

To compute the disturbance torque momentum, the total disturbance torques, XDT, YDT, and ZDT, are first calculated by summing the atmospheric, solar, and gravity gradient torques by

$$\text{XDT} = \text{TAX} + \text{TSX} + \text{TGX}$$

$$\text{YDT} = \text{TAY} + \text{TSY} + \text{TGY}$$

$$\text{ZDT} = \text{TAZ} + \text{TSZ} + \text{TGZ}$$

Using these totals, the disturbance torques, HTX, HTY, and HTZ, are then computed using

$$HTX = XDT\ (TL)(T)$$

$$HTY = YDT\ (TL)(T)$$

$$HTZ = ZDT\ (TL)(T)$$

To compute the orientation tracking momentums, HTRAKX, HTRAKY, and HTRAKZ, the program first determines whether the spacecraft is solar oriented, inertial oriented, or Earth oriented. If the spacecraft is either solar or inertially oriented, there is no orientation tracking momentum and the following assignments are made:

$$HTRAKX = 0.0$$

$$HTRAKY = 0.0$$

$$HTRAKZ = 0.0$$

If the spacecraft is Earth oriented, the program will determine whether its body axis is sideways, nose down, or nose forward with respect to the orbital velocity vector (fig. 2). The following assignments are made if its body axis is

| | |
|---|---|
| (1) Sideways: | $HTRAKX = XJ(\dot{\theta})$<br>$HTRAKY = 0.0$<br>$HTRAKZ = 0.0$ |
| (2) Nose down: | $HTRAKX = 0.0$<br>$HTRAKY = YJ(\dot{\theta})$<br>$HTRAKZ = 0.0$ |
| (3) Nose forward: | $HTRAKZ = 0.0$<br>$HTRAKY = YJ(\dot{\theta})$<br>$HTRAKZ = 0.0$ |

The total momentum in each axis is computed by summing the maneuver, disturbance torque, and orientation tracking momentums

$$HX = HXMAN + HTX + HTRAKX$$

$$HY = HYMAN + HTY + HTRAKY$$

$$HZ = HZMAN + HTZ + HTRAKZ$$

Control Device's Momentum Absorption Requirements

The momentum absorption requirements for the reaction wheel, CMG, and DMCD are calculated to determine the wheel weight. The maximum momentum in any axis, HMAX, is used for the calculation of nongimbled reaction wheel weight. Using an intrinsic FORTRAN function, HMAX is determined by

$$HMAX = AMAX1\ (HX,\ HY,\ HZ)$$

Here HMAX will be assigned the largest value between HX, HY, and HZ. To compute the CMG momentum for computing wheel weight, the minimum momentum in any axis, HMIN, is first calculated using another intrinsic FORTRAN function

$$HMIN = AMIN1\ (HX,\ HY,\ HZ)$$

Here HMIN will be assigned the smallest value between HX, HY, and HZ. From this, the CMG slew angle, GAMMA, is calculated from

$$GAMMA = TAN^{-1} \left[ \frac{(XNNN-2)\ HMIN}{(XNNN)\ HMAX} \right]$$

Then using the preceding HMIN and GAMMA calculations, the CMG wheel momentum, HCMG, is calculated using

$$HCMG = \frac{HMIN}{(XNNN)\ \sin\ (GAMMA)}$$

The peak gimble rate, DELDOT, is computed by

$$DELDOT = \frac{HCMG}{57.3\ (TACCEL)}$$

and the peak torquer torque, TCMG, by

$$\mathrm{TCMG} = \frac{(\mathrm{HCMG})\ \mathrm{AMAX1}\ (\mathrm{PDOTRX},\ \mathrm{PDOTRY},\ \mathrm{PDOTRZ})}{57.3}$$

To compute the DMCD momentum for computing wheel weight, the spin and transverse axis momentum absorption requirements must be calculated. The DMCD configuration for both the spin and transverse axes is shown in figure 4.

The spin axis absorption momentum, DELTHU, is calculated by

$$\mathrm{DELTHU} = \frac{\mathrm{HX}}{2}$$

and the transverse axis absorption momentum by

$$\mathrm{HU} = \frac{\mathrm{AMAX1}\ (\mathrm{HY},\ \mathrm{HZ})}{2\left(\frac{\mathrm{UPSILN}}{2}\right)}$$

From these calculations, the total DMCD wheel momentum, HTDMCD, is computed by summing the spin and transverse axes absorption momentums

$$\mathrm{HTDMCD} = \mathrm{DELTHU} + \mathrm{HU}$$

## Wheel Weight

The final step of the program is to calculate the wheel weight. The wheel mass is calculated from the relationship between the rotational inertia of the wheel and its radius,

$$I = MR^2$$

Since the angular momentum, L, is equal to the rotational inertia of the wheel, I, multiplied by its angular velocity (wheel speed),

$$L = I\omega$$

Then

$$L = MR^2\omega$$

and

$$M = \frac{L}{R^2\omega}$$

For a solid cylinder:

$$I = \frac{MR^2}{2}$$

and

$$L = \frac{MR^2}{2}\,\omega$$

and

$$M = \frac{2L}{R^2\omega}$$

Finally, for an annular cylinder

$$I = M\,\frac{\left(R_1{}^2 + R_2{}^2\right)}{2}$$

and

$$L = M\,\frac{\left(R_1{}^2 + R_2{}^2\right)\omega}{2}$$

and

$$M = \frac{2L}{\omega\left(R_1{}^2 + R_2{}^2\right)}$$

where $R_1$ is the wheel's inner radius and $R_2$ is the outer radius. The acceleration of gravity, ACC, at the spacecraft's altitude is then calculated by

$$ACC = \frac{u}{RP^2}$$

The wheel masses are next multiplied by the acceleration of gravity to determine the wheel weight at the spacecraft's altitude. Thus, the equations used in computing wheel weight are

(1) For a hoop $$W = \frac{L\ (ACC)}{R^2 \omega}$$

(2) For a solid cylinder $$W = \frac{2L\ (ACC)}{R^2 \omega}$$

(3) For an annular cylinder $$W = \frac{2L\ (ACC)}{\omega\left(R_1^{\ 2} + R_2^{\ 2}\right)}$$

## Trade-Off and Output

Calculated wheel weight curves are shown in figures 5 through 7 for wheel radii varying from 1 inch (2.54 cm) to 10 inches (25.4 cm) and wheel speeds of 500 to 5000 rpm's in increments of 500 rpm's to illustrate the trade-off potential between weight, radius, and wheel speed. Weights of hoop, solid cylinder, and annular cylinder momentum wheels are presented for the nongimbled, CMG and DMCD systems. Calculations were made in U.S. Customary Units. Conversion factors for values used in this report are given in table 1. The program's input default values, which were used to generate the weight curves (figs. 5 to 7), are given in figure 8.

In addition to the weight curves, values of 22 calculated parameters are outputted, including the maximum torque in any axis which is calculated by

$$TMAX = \frac{AMAX1\ (HXMAN,\ HYMAN,\ HZMAN)}{TACCEL + AMAX1\ (XDT,\ YDT,\ ZDT)}$$

These other calculated parameters are

RP - Radius at periapsis

$\dot{\theta}$ - Maximum orbital rate of change

T - Orbital period

$H \begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix} MAN$ - X-, Y-, Z-axis maneuver momentum

$HT \begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix}$ - X-, Y-, Z-axis disturbance torque momentum

$HTRAK \begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix}$ - X-, Y-, Z-axis orientation tracking momentum

$H \begin{Bmatrix} X \\ Y \\ Z \end{Bmatrix}$ - X-, Y-, Z-axis total momentum

HMAX - Maximum momentum, any axis

HCMG - CMG wheel momentum

HTDMCD - DMCD wheel momentum

DELDOT - Peak gimble rate

TCMG - Peak torquer torque

TMAX - Maximum torque in any axis

Calculated values of the output parameters are given in figure 9 for the weight curves illustrated.

## CONCLUDING REMARKS

An interactive computer program has been developed at the Langley Research Center to size momentum wheels for various momentum storage and control devices for orbiting spacecraft. The program considers hoop, solid cylinder, and annular cylinder wheels.

Wheel weights are calculated and are shown for a series of wheel radii and wheel rotational speeds. Intermediate calculated parameters are also presented.

APPENDIX

MOMENTUM WHEEL SIZING PROGRAM

The momentum wheel sizing program listing is given in this appendix.

```
      PROGRAM WHEEL(INPUT,OUTPUT,TAPE1,TAPE2,TAPE3)
      REAL N
      COMMON INPUT,LOC,REAL,EXP
      COMMON /WEIR/ WEIGH1,WEIGH2,WEIGH3,WEIGH4,WEIGH5,
     *WEIGH6,WEIGH7,WEIGH8,WEIGH9,WEIGH0,R
      COMMON /URPI/U,RP,PI
      DIMENSION R(122),WEIGH1(122),WEIGH2(122)
      DIMENSION WEIGH3(122),WEIGH4(122),WEIGH5(122)
      DIMENSION WEIGH6(122),WEIGH7(122),WEIGH8(122)
      DIMENSION WEIGH9(122),WEIGH0(122)
      DIMENSION LOC(5),REAL(9),EXP(12),INPUT(277)
C        RADIUS OF EARTH
      DATA RE/3441.66/
C        EARTH'S GRAVITATIONAL CONSTANT
      DATA U/1.407850464E16/
      DATA PI/3.141592654/
      DATA WEIGH1(1),WEIGH2(1),WEIGH3(1),WEIGH4(1),WEIGH5(1),
     *WEIGH6(1),WEIGH7(1),WEIGH8(1),WEIGH9(1),WEIGH0(1),
     *R(1)/11*101./
      CALL INITT(120)
      CALL FEET(RE)
 20   PRINT *,"ENTER 1 FOR DEFAULT VALUES"
      PRINT *,"ENTER 2 FOR VALUES FROM LAST RUN"
      PRINT *,"ENTER 3 FOR PERMANENTLY SAVED VALUES"
      PRINT *,"ENTER 4 TO STOP"
      READ *,IVALUE
      PRINT 5025
      IF (IVALUE.EQ.1) GOTO 40
      IF (IVALUE.EQ.2) GOTO 60
      IF (IVALUE.EQ.3) GOTO 70
      IF (IVALUE.EQ.4) GOTO 999
      GOTO 20
 40   READ (1,5000) (LOC(J),J=1,3)
      READ (1,5005) (REAL(J),J=1,9)
      READ (1,5010) (EXP(J),J=1,12)
      READ (1,5015) (INPUT(J),J=1,277)
      REWIND 1
      CALL REPLACE
      GOTO 80
 60   READ (2,5000) (LOC(J),J=1,3)
      READ (2,5005) (REAL(J),J=1,9)
      READ (2,5010) (EXP(J),J=1,12)
      READ (2,5015) (INPUT(J),J=1,277)
      REWIND 2
```

```
      CALL REPLACE
      GOTO 80
 70   READ (3,5000) (LOC(J),J=1,3)
      READ (3,5005) (REAL(J),J=1,9)
      READ (3,5010) (EXP(J),J=1,12)
      READ (3,5015) (INPUT(J),J=1,277)
      REWIND 3
      CALL REPLACE
 80   INOSE= LOC(1)
      ISATOR= LOC(2)
      XNNN= LOC(3)
      APOA= REAL(1)
      PER= REAL(2)
      ORBINC= REAL(3)
      TL= REAL(4)
      TACCEL= REAL(5)
      PDOTRX= REAL(6)
      PDOTRY= REAL(7)
      PDOTRZ= REAL(8)
      UPSILN= REAL(9)
      TAX= EXP(1)
      TGX= EXP(2)
      TSX= EXP(3)
      TAY= EXP(4)
      TGY= EXP(5)
      TSY= EXP(6)
      TAZ= EXP(7)
      TGZ= EXP(8)
      TSZ= EXP(9)
      XJ= EXP(10)
      YJ= EXP(11)
      ZJ= EXP(12)
      WRITE (2,5000) (LOC(J),J=1,3)
      WRITE (2,5005) (REAL(J),J=1,9)
      WRITE (2,5010) (EXP(J),J=1,12)
      WRITE (2,5015) (INPUT(J),J=1,277)
      REWIND 2
      PRINT *,"DO YOU WANT TO SAVE NEW VALUES PERMANENTLY?"
      PRINT *,"ENTER 1 FOR YES"
      PRINT *,"ENTER 2 FOR NO"
      READ *,JOE
      PRINT 5040
      IF (JOE.EQ.1) 90,95
 90   WRITE (3,5000) (LOC(J),J=1,3)
      WRITE (3,5005) (REAL(J),J=1,9)
      WRITE (3,5010) (EXP(J),J=1,12)
      WRITE (3,5015) (INPUT(J),J=1,277)
      REWIND 3
 95   CALL FEET(APOA)
      CALL FEET(PER)
```

```
C         COMPUTE SEMIMAJOR AXIS OF ELIPSE
      A= (2.*RE+APOA+PER)*0.5
C         COMPUTE ORBITAL PERIOD
      T= 2.*PI*A**(1.5)/SQRT(U)
C         COMPUTE RADIUS OF PERIAPSIS
      RP= RE+PER
C         COMPUTE FOCUS OF ELIPSE
      C= A-RP
C         COMPUTE ECCENTRICITY
      ECC= C/A
C         COMPUTE MEAN ORBITAL MOTION
      N= SQRT(U/A**3)
C        COMPUTE ECCENTRIC ANOMALLY
      E= N+ECC*SIN(N)+0.5*ECC**2*SIN(2*N)
C         COMPUTE MAX ORBITAL ANGULAR RATE
      THETAD= (N*SQRT(1.-ECC**2))/((1.-ECC*COS(E))**2)
C         COMPUTE MOMENTUM REQUIREMENTS FOR SIZING NON-GIMBALLED
C         MOMONTUM WHEELS
C         MANEUVER MOMENTUM
      HXMAN= XJ*PDOTRX/57.3
      HYMAN= YJ*PDOTRY/57.3
      HZMAN= ZJ*PDOTRZ/57.3
C         COMPUTE DISTURBANCE TORQUES
      XDT= TAX+TGX+TSX
      YDT= TAY+TGY+TSY
      ZDT= TAZ+TGZ+TSZ
C         COMPUTE DISTURBANCE TORQUE MOMENTUM
      HTX= XDT*TL*T
      HTY= YDT*TL*T
      HTZ= ZDT*TL*T
C         COMPUTE ORIENTATION TRACKING MOMENTUM REQUIREMENTS
      GOTO (120,100,100) ISATOR
C         INERTIAL AND SOLAR ORIENTATION
100   HTRAKX= 0.0
      HTRAKY= 0.0
      HTRAKZ= 0.0
      GOTO 200
C         EARTH ORIENATION
120   GOTO (140,160,180) INOSE
C         SIDEWAYS
140   HTRAKX= XJ*THETAD
      HTRAKY= 0.0
      HTRAKZ= 0.0
      GOTO 200
C         NOSE DOWN
160   HTRAKX= 0.0
      HTRAKY= YJ*THETAD
      HTRAKZ= 0.0
      GOTO 200
C         NOSE FORWARD
```

```
180   HTRAKX= 0.0
      HTRAKY= YJ*THETAD
      HTRAKZ= 0.0
C         COMPUTE TOTAL MOMENTUM PER ORBIT
200   HX= (HXMAN+HTX+HTRAKX)
      HY= (HYMAN+HTY+HTRAKY)
      HZ= (HZMAN+HTZ+HTRAKZ)
C         COMPUTE MAX MOMENTUM ANY AXIS
      HMAX= AMAX1(HX,HY,HZ)
C         COMPUTE MAX TORQUE ANY AXIS
      TMAX= AMAX1(HXMAN,HYMAN,HZMAN)/TACCEL+AMAX1(XDT,YDT,ZDT)
C        COMPUTE MINIMUM MOMENTUM
      HMIN= AMIN1(HX,HY,HZ)
C        COMPUTE SKEW ANGLE
      GAMMA= ATAN((XNNN-2.)*HMIN/(XNNN*HMAX))
C        COMPUTE CMG WHEEL MOMENTUM
      HCMG= HMIN/(XNNN*SIN(GAMMA))
C        COMPUTE PEAK GIMBLE RATE
      DELDOT= HCMG/TACCEL*57.3
C        COMPUTE PEAK TORQUER TORQUE
      TCMG= HCMG*AMAX1(PDOTRX,PDOTRY,PDOTRZ)/57.3
C        COMPUTE SPIN AXIS ABSORPTION REQUIREMENT
      DELTHU= HX/2.
C        COMPUTE TRANSVERSE AXIS ABSORPTION REQUIREMENT
      HU= AMAX1(HY,HZ)/(2.*UPSILN/57.3)
C        COMPUTE TOTAL DMCD WHEEL MOMENTUM
      HTDMCD= DELTHU+HU
      PRINT 5030,(INPUT(J),J=163,167),APOA
      PRINT 5030,(INPUT(J),J=168,172),RP
      PRINT 5030,(INPUT(J),J=173,177),THETAD
      PRINT 5030,(INPUT(J),J=178,182),T
      PRINT 5025
      PRINT 5035,(INPUT(J),J=273,277)
      PRINT 5025
      PRINT 5030,(INPUT(J),J=183,187),HXMAN
      PRINT 5030,(INPUT(J),J=188,192),HYMAN
      PRINT 5030,(INPUT(J),J=193,197),HZMAN
      PRINT 5025
      PRINT 5030,(INPUT(J),J=198,202),HTX
      PRINT 5030,(INPUT(J),J=203,207),HTY
      PRINT 5030,(INPUT(J),J=208,212),HTZ
      PRINT 5025
      PRINT 5030,(INPUT(J),J=213,217),HTRAKX
      PRINT 5030,(INPUT(J),J=218,222),HTRAKY
      PRINT 5030,(INPUT(J),J=223,227),HTRAKZ
      PRINT 5025
      PRINT 5030,(INPUT(J),J=228,232),HX
      PRINT 5030,(INPUT(J),J=233,237),HY
      PRINT 5030,(INPUT(J),J=238,242),HZ
      PRINT 5025
```

```
      PRINT 5030,(INPUT(J),J=243,247),HMAX
      PRINT 5030,(INPUT(J),J=248,252),TMAX
      PRINT 5030,(INPUT(J),J=253,257),HCMG
      PRINT 5025
      PRINT 5030,(INPUT(J),J=258,262),DELDOT
      PRINT 5030,(INPUT(J),J=263,267),TCMG
      PRINT 5030,(INPUT(J),J=268,272),HTDMCD
      PRINT 5040
      REWIND 3
220   PRINT *,"DO YOU WANT WHEEL SPEED, SIZE AND WEIGHT PLOT?"
      PRINT *,"ENTER 1 FOR NON-GIMBLED PLOT"
      PRINT *,"ENTER 2 FOR CMG PLOT"
      PRINT *,"ENTER 3 FOR DMCD PLOT"
      PRINT *,"ENTER 4 FOR NO PLOT"
      READ *,IPLOT
      IF (IPLOT.LT.1.OR.IPLOT.GT.4) GOTO 220
      GOTO (240,260,280,20) IPLOT
240   CALL STUFF(HMAX,1)
      GOTO 220
260   CALL STUFF(HCMG,2)
      GOTO 220
280   CALL STUFF(HTDMCD,3)
      GOTO 220
999   CALL FINITT(0,700)
5000  FORMAT (5I5)
5005  FORMAT (5F10.3)
5010  FORMAT (5E10.4)
5015  FORMAT (9(4A10,A5/),9(5A10/),12(6A10/),23(5A10/))
5020  FORMAT (5F5.2)
5025  FORMAT (1X)
5030  FORMAT (5A10,E14.7)
5035  FORMAT (6A10)
5040  FORMAT (///)
      STOP
      END
C
C        SUBROUTINE USED TO CHANGE PLOT AXES VALUES
C
      SUBROUTINE CHANGE(XVALUE,YVALUE,XMIN,XMAX,YMIN,YMAX,JJ)
      INTEGER XVALUE,YVALUE
5     PRINT *,"DO YOU WANT NEW GRAPH?"
      PRINT *,"IF YES ENTER 1"
      PRINT *,"IF NO ENTER 2"
      READ *,JJ
      IF (JJ.EQ.1) 20,10
10    IF (JJ.EQ.2) 120,5
20    PRINT *,"DO YOU WANT TO CHANGE AXIS MAX AND MIN VALUES"
      PRINT *,"IF YES ENTER 1"
      PRINT *,"IF NO ENTER 2"
      READ *,II
```

```
      IF (II.EQ.1) 40,80
40    PRINT *,"ENTER  X-AXIS MIN AND MAX VALUE"
      READ *,XMIN,XMAX
      IF (XMIN.EQ.0.0.OR.XMIN.GE.XMAX) GOTO 40
      IF (XMIN.NE.1.0.OR.XMAX.NE.10.0) JJ=1
60    PRINT *,"ENTER Y-AXIS MIN AND MAX VALUE"
      READ *,YMIN,YMAX
      IF (YMIN.GE.YMAX) GOTO 60
80    PRINT *,"DO YOU WANT TO CHANGE # OF INTERVALS"
      PRINT *,"IF YES ENTER 1"
      PRINT *,"IF NO ENTER 2"
      READ *,KK
      IF (KK.EQ.1) 100,120
100   PRINT *,"ENTER # OF X-AXIS INTERVALS"
      READ *,XVALUE
      PRINT *,"ENTER # OF Y-AXIS INTERVALS"
      READ *,YVALUE
120   RETURN
      END
C
C        SUBROUTINE USED TO PLOT CURVES
C
      SUBROUTINE STUFF (ANGMOM,JJJ)
      INTEGER RPM,XVALUE,YVALUE
      DIMENSION WEIGH1(122),WEIGH2(122),WEIGH3(122),WEIGH4(122)
      DIMENSION WEIGH6(122),WEIGH7(122),WEIGH8(122),WEIGH9(122)
      DIMENSION WEIGH0(122),R(122),WEIGH5(122),R1(122)
      DIMENSION IARRAY(10)
      COMMON /WEIR/ WEIGH1,WEIGH2,WEIGH3,WEIGH4,WEIGH5,WEIGH6,
     *WEIGH7,WEIGH8,WEIGH9,WEIGH0,R
      COMMON /URPI/ U,RP,PI
      XMIN= 1.0
      XMAX= 10.0
      YMIN= 0.0
      YMAX= 5.0
      XVALUE= 9
      YVALUE= 5
20    PRINT *,"DO YOU WANT WHEEL PLOT FOR:"
      PRINT *,"   1- HOOP"
      PRINT *,"   2- SOLID CYLINDER"
      PRINT *,"   3- ANNULAR CYLINDER"
      READ *,M
      IF (M.LT.1.OR.M.GT.3) 20,40
40    IF (M.EQ.3) 60,80
60    PRINT *,"ENTER THICKNESS OF ANNULAR CYLINDER IN INCHES"
      READ *,THICK
80    I=1
      ZNUM= (XMAX-XMIN)/100.
      RADIUS= XMIN
```

APPENDIX

```
      IF (M.EQ.3) 100,140
100   DO 120 J=2,102
      R(J)= RADIUS
      R1(J)= RADIUS+THICK
      RADIUS= RADIUS+ZNUM
120   CONTINUE
      GOTO 180
140   DO 160 J=2,102
      R(J)= RADIUS
      R1(J)= 0.0
      RADIUS= RADIUS+ZNUM
160   CONTINUE
180   ACC= U/RP**2
      IF (M.EQ.1) 200,220
200   HUMACC= ANGMOM*ACC
      GOTO 240
220   HUMACC= 2.*ANGMOM*ACC
240   DO 480 RPM= 500,5000,500
      OMEGA= RPM/60.*2.*PI
      DO 460 J=2,102
      GOTO (260,280,300,320,340,360,380,400,420,440) I
260   WEIGH1(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
280   WEIGH2(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
300   WEIGH3(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
320   WEIGH4(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
340   WEIGH5(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
360   WEIGH6(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
380   WEIGH7(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
400   WEIGH8(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
420   WEIGH9(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
      GOTO 460
440   WEIGH0(J)= HUMACC/(OMEGA*(R(J)**2+R1(J)**2))
460   CONTINUE
      I= I+1
480   CONTINUE
490   CALL BINITT
      CALL SLIMY(155,730)
      CALL SLIMX(170,920)
      CALL XTICS (XVALUE)
      CALL YTICS (YVALUE)
      CALL XNEAT (0)
      CALL YNEAT (0)
```

```
      CALL NEWPAG
      CALL NOTX (5,500,5,5HWHEEL)
      CALL NOTX(0,485,6,6HWEIGHT)
      CALL NOTX(8,470,4,4H(LB))
      CALL NOTX (440,60,17,17HWHEEL RADIUS (IN))
      CALL NOTX (370,0,36,36HRPM'S VARY FROM 500 TO 5000 STEP 500)
      GOTO (500,520,540) JJJ
500   CALL NOTX (445,30,16,16HNON-GIMBLED PLOT)
       GOTO 560
520   CALL NOTX (480,30,8,8HCMG PLOT)
      GOTO 560
540   CALL NOTX (470,30,9,9HDMCD PLOT)
560   GOTO (580,600,620) N
580   CALL NOTX (490,15,4,4HHOOP)
      GOTO 640
600   CALL NOTX (450,15,14,14HSOLID CYLINDER)
      GOTO 640
620   CALL NOTX (350,15,28,28HANNULAR CYLINDER: THICKNESS-)
      CALL NOTX (650,15,6,6HINCHES)
      CALL FFORM (THICK,5,2,IARRAY,32)
      CALL NOTAT (600,15,5,IARRAY)
640   CALL DLIMX (XMIN,XMAX)
      CALL DLIMY (YMIN,YMAX)
      CALL CHECK (R,WEIGH1)
      CALL DSPLAY (R,WEIGH1)
      CALL CPLOT (R,WEIGH2)
      CALL CPLOT (R,WEIGH3)
      CALL CPLOT (R,WEIGH4)
      CALL CPLOT (R,WEIGH5)
      CALL CPLOT (R,WEIGH6)
      CALL CPLOT (R,WEIGH7)
      CALL CPLOT (R,WEIGH8)
      CALL CPLOT (R,WEIGH9)
      CALL CPLOT (R,WEIGH0)
      CALL HOME
      CALL EPAUSE
      CALL CHANGE (XVALUE,YVALUE,XMIN,XMAX,YMIN,YMAX,JJ)
      IF (JJ.EQ.2) RETURN
      IF (JJ.EQ.3) GOTO 490
      GOTO 80
      END
C
C        SUBROUTINE USED TO CHANGE NAUTICAL MILES TO FEET
C
      SUBROUTINE FEET (CHANGE)
      CHANGE= CHANGE*6076.103
      RETURN
      END
C
```

APPENDIX

```
C        SUBROUTINE USED TO DISPLAY INPUTS
C
      SUBROUTINE  REPLACE
      COMMON INPUT,LOC,REAL,EXP
      DIMENSION LOC(5),REAL(9),EXP(12),INPUT(277)
20    PRINT 5000,INPUT(1),INPUT(2),INPUT(3),
     *INPUT(4),INPUT(5),LOC(1)
      DO 40 J=6,20,5
      PRINT 5015,INPUT(J),INPUT(J+1),INPUT(J+2),
     *INPUT(J+3),INPUT(J+4)
40    CONTINUE
      PRINT 5005,INPUT(21),INPUT(22),INPUT(23),
     *INPUT(24),INPUT(25),LOC(2)
      DO 60 J=26,40,5
      PRINT 5015,INPUT(J),INPUT(J+1),INPUT(J+2),
     *INPUT(J+3),INPUT(J+4)
60    CONTINUE
      PRINT 5010,INPUT(41),INPUT(42),INPUT(43),
     *INPUT(44),INPUT(45),LOC(3)
80    PRINT 5020
      READ *,CHECK
      PRINT 5035
      IF (CHECK.EQ.1) GOTO 120
      IF (CHECK.EQ.2) 100,80
100   CALL ONE
      PRINT 5035
      GOTO 20
120   II= 46
      DO 140 J=1,9
      PRINT 5025,J,INPUT(II),INPUT(II+1),INPUT(II+2),
     *INPUT(II+3),INPUT(II+4),REAL(J)
      II= II+5
140   CONTINUE
160   PRINT 5020
      READ *,CHECK
      PRINT 5035
      IF (CHECK.EQ.1) GOTO 200
      IF (CHECK.EQ.2) 180,160
180   CALL TWO
      PRINT 5035
      GOTO 120
200   II= 91
      DO 220 J=1,12
      PRINT 5030,J,INPUT(II),INPUT(II+1),INPUT(II+2),
     *INPUT(II+3),INPUT(II+4),INPUT(II+5),EXP(J)
      II= II+6
220   CONTINUE
240   PRINT 5020
      READ *,CHECK
      PRINT 5035
```

```
      IF (CHECK.EQ.1) GOTO 280
      IF (CHECK.EQ.2) 260,240
260   CALL THREE
      PRINT 5035
      GOTO 200
280   RETURN
5000  FORMAT (1X,"1",1X,4A10,A5,I3)
5005  FORMAT (1X,"2",1X,4A10,A5,I3)
5010  FORMAT (1X,"3",1X,4A10,A5,I3)
5015  FORMAT (3X,4A10,A5)
5020  FORMAT (///,1X,"IF INPUTS OK ENTER 1",/,1X,"IF WANT TO CHANGE",
     *" ENTER 2")
5025  FORMAT (I2,1X,5A10,F10.3)
5030  FORMAT (I2,1X,6A10,E10.4)
5035  FORMAT (//)
      END
C
C        SUBROUTINE USED TO CHANGE INTEGER INPUTS
C
      SUBROUTINE  ONE
      COMMON INPUT,LOC,REAL,EXP
      DIMENSION LOC(5),REAL(9),EXP(12),INPUT(277)
20    PRINT *,"ENTER NUMBER, NEW VALUE (INTEGER)"
      PRINT *,"ENTER 0,0 TO STOP"
40    READ *, NUM,IVALUE
      IF (NUM.EQ.0) RETURN
      IF ((NUM.EQ.1.OR.NUM.EQ.2).AND.(IVALUE.LT.1.OR.IVALUE.GT.3))
     *GOTO 20
      IF (NUM.LT.1.OR.NUM.GT.3) GOTO 20
      LOC(NUM)= IVALUE
      GOTO 40
      RETURN
      END
C
C        SUBROUTINE USED TO CHANGE REAL INPUTS
C
      SUBROUTINE  TWO
      COMMON INPUT,LOC,REAL,EXP
      DIMENSION LOC(5),REAL(9),EXP(12),INPUT(277)
20    PRINT *,"ENTER NUMBER, NEW VALUE (REAL)"
      PRINT *,"ENTER 0,0. TO STOP"
40    READ *,NUM,VALUE
      IF (NUM.EQ.0) RETURN
      IF (NUM.LT.1.OR.NUM.GT.9) GOTO 20
      REAL(NUM)= VALUE
      GOTO 40
      RETURN
      END
C
```

APPENDIX

```
C        SUBROUTINE USED TO CHANGE EXP. INPUTS
C
      SUBROUTINE  THREE
      COMMON INPUT,LOC,REAL,EXP
      DIMENSION LOC(5),REAL(9),EXP(12),INPUT(277)
20    PRINT *,"ENTER NUMBER, NEW VALUE (EXP)"
      PRINT *,"ENTER 0,0. TO STOP"
40    READ *,NUM,VALUE
      IF (NUM.EQ.0) RETURN
      IF (NUM.LT.1.OR.NUM.GT.12) GOTO 20
      EXP(NUM)= VALUE
      GOTO 40
      RETURN
      END
/
```

REFERENCES

1. Systems Cost/Performance Analysis (Study 2.3), Final Report; 31 March 1975; Prepared by The Aerospace Corporation, Los Angeles, California. Prepared for Office of Manned Flight, National Aeronautics and Space Administration, Washington, DC, 20546, under Contract NASW-2727.

2. Large Advanced Space Systems Computer Program, Monthly Progress Reports (Final Report Pending); Prepared by General Dynamics Convair Division, San Diego, California, 92138. Prepared for Langley Research Center, Hampton, Virginia, 23665, under Contract NAS1-15462.

3. Space Flight Handbooks, Volume 1, Orbital Flight Handbook, Part 1 - Basic Techniques and Data. Prepared for the George C. Marshall Space Flight Center, Huntsville, Alabama; produced by the Martin Company under Contract NAS8-5031; NASA SP 33, Part 1, 1963.

TABLE I. CONVERSION FACTORS FROM U.S. CUSTOMARY TO S.I. UNITS

| | | |
|---|---|---|
| 1 inch | = | 2.54 centimeters |
| 1 foot | = | 0.305 meters |
| 1 mile | = | 1.61 kilometers |
| 1 pound (wt) | = | 4.95 newtons = 0.454 kilograms |
| 1 foot-pound | = | 1.38 newton-meters |
| 1 slug-foot$^2$ | = | 1.36 kilogram-meters$^2$ |
| 1 $\frac{\text{slug-foot}^2}{\text{seconds}}$ | = | 1.36 $\frac{\text{kilogram-meters}^2}{\text{seconds}}$ |




Figure 1. Flow diagram.




Figure 2. Relationship between the orbital reference frame and the three vehicle reference frames.




LEGEND:
a = Semimajor axis of orbit ellipse
A = Apoapsis altitude
C = Focus of orbit ellipse
P = Periapsis altitude
R = Radius of Earth
RP = Radius of periapsis

Figure 3. Orbit definition.




Figure 4. DMCD configuration.




RPM'S VARY FROM 500 TO 5000 STEP 500

(a) Hoop.

Figure 5. Weight curves for nongimbled reaction wheel.




RPM'S VARY FROM 500 TO 5000 STEP 500

(b) Solid cylinder.

Figure 5. Continued.




(c) Annular cylinder.

Figure 5. Concluded.




RPM'S VARY FROM [illegible] TO [illegible] STEP [illegible]

(a) Hoop.

Figure 6. Weight curve for CMG.




RPM'S VARY FROM 500 TO 5000 STEP 500

(b) Solid cylinder.

Figure 6. Continued.




(c) Annular cylinder.

Figure 6. Concluded.




RPM'S VARY FROM 500 TO 5000 STEP 500

(a) Hoop.

Figure 7. Weight curves for DMCD.




RPM'S VARY FROM 500 TO 5000 STEP 500

(b) Solid cylinder.

Figure 7. Continued.




(c) Annular cylinder.

Figure 7. Concluded.

| | | | |
|---|---|---|---|
| 1 | INOSE- | VEHICLE POINTING ORIENTATION<br>1= SIDEWAYS<br>2= NOSE DOWN<br>3= NOSE FORWARD | 3 |
| 2 | ISATOR- | CELESTIAL ORIENTATION<br>1= EARTH<br>2= SUN<br>3= INERTIAL | 1 |
| 3 | XNNN- | NUMBER OF GIMBALED CONFIGURATIONS | 4 |

IF INPUTS OK ENTER 1
IF WANT TO CHANGE ENTER 2

| | | | |
|---|---|---|---|
| 1 | APOA- | LENGTH OF APOAPSIS (ML) | 175.000 |
| 2 | PER- | LENGTH OF PERIAPSIS (ML) | 100.000 |
| 3 | ORBINC- | ORBIT INCLINATION (DEG) | 57.300 |
| 4 | TL- | TIME BETWEEN WHEEL UNLOADING (ORBITS) | 1.000 |
| 5 | TACCEL- | MANEUVER ACCELERATION TIME (SEC) | 20.000 |
| 6 | PDOTRX- | X-AXIS MANEUVER RATE (DEG/SEC) | 1.000 |
| 7 | PDOTRY- | Y-AXIS MANEUVER RATE (DEG/SEC) | 1.000 |
| 8 | PDOTRZ- | Z-AXIS MANEUVER RATE (DEG/SEC) | 1.000 |
| 9 | UPSILN- | ANGULAR PIVOTING FOR DMCD (DEG) | 20.000 |

IF INPUTS OK ENTER 1
IF WANT TO CHANGE ENTER 2

| | | | |
|---|---|---|---|
| 1 | TAX- | X-AXIS ATMOSPHERIC DIST. TORQUE (FT-LBS) | 0. |
| 2 | TGX- | X-AXIS GRAVITY GRAD. DIST. TORQUE (FT-LBS) | .2859E-10 |
| 3 | TSX- | X-AXIS SOLAR DIST. TORQUE (FT-LBS) | 0. |
| 4 | TAY- | Y-AXIS ATMOSPHERIC DIST. TORQUE (FT-LBS) | .8541E-06 |
| 5 | TGY- | Y-AXIS GRAVITY GRAD. DIST. TORQUE (FT-LBS) | .3791E-10 |
| 6 | TSY- | Y-AXIS SOLAR DIST. TORQUE (FT-LBS) | 0. |
| 7 | TAZ- | Z-AXIS ATMOSPHERIC DIST. TORQUE (FT-LBS) | .8477E-07 |
| 8 | TGZ- | Z-AXIS GRAVITY GRAD. DIST. TORQUE (FT-LBS) | 0. |
| 9 | TSZ- | Z-AXIS SOLAR DIST. TORQUE (FT-LBS) | 0. |
| 10 | XJ- | X-AXIS SPACECRAFT ROTATIONAL INERTIA (SLUG-FT**2) | .1595E+04 |
| 11 | YJ- | Y-AXIS SPACECRAFT ROTATIONAL INERTIA (SLUG-FT**2) | .1380E+04 |
| 12 | ZJ- | Z-AXIS SPACECRAFT ROTATIONAL INERTIA (SLUG-FT**2) | .8034E+03 |

IF INPUTS OK ENTER 1
IF WANT TO CHANGE ENTER 2

Figure 8. Input default values.

| | | |
|---|---|---|
| APOA- | APOAPSIS ALTITUDE (FT) | .1063318E+07 |
| RP- | RADIUS AT PERIAPSIS (FT) | .2151949E+08 |
| THETAD- | MAXIMUM ORBITAL ANGULAR RATE (RAD/SEC) | .1194795E-02 |
| T- | ORBITAL PERIOD (SEC) | .5370454E+04 |

ALL MOMENTUM IN (SLUG-FT**2)/SEC

| | | |
|---|---|---|
| HXMAN- | X-AXIS MANEUVER MOMENTUM | .2783595E+02 |
| HYMAN- | Y-AXIS MANEUVER MOMENTUM | .2408377E+02 |
| HZMAN- | Z-AXIS MANEUVER MOMENTUM | .1402094E+02 |
| HTX- | X-AXIS DISTURBANCE TORQUE MOMENTUM | .1535413E-06 |
| HTY- | Y-AXIS DISTURBANCE TORQUE MOMENTUM | .4587108E-02 |
| HTZ- | Z-AXIS DISTURBANCE TORQUE MOMENTUM | .4552533E-03 |
| HTRAKX- | X-AXIS ORIENTATION TRACKING MOMENTUM | 0. |
| HTRAKY- | Y-AXIS ORIENTATION TRACKING MOMENTUM | .1648818E+01 |
| HTRAKZ- | Z-AXIS ORIENTATION TRACKING MOMENTUM | 0. |
| HX- | TOTAL X-AXIS MOMENTUM | .2783595E+02 |
| HY- | TOTAL Y-AXIS MOMENTUM | .2573717E+02 |
| HZ- | TOTAL Z-AXIS MOMENTUM | .1402140E+02 |
| HMAX- | MAXIMUM MOMENTUM ANY AXIS | .2783595E+02 |
| TMAX- | MAXIMUM TORQUE ANY AXIS (FT-LBS) | .1391798E+01 |
| HCMG- | CMG WHEEL MOMENTUM | .1435261E+02 |
| DELDOT- | PEAK GIMBAL RATE (RAD/SEC) | .4112024E+02 |
| TCMG- | PEAK TORQUER TORQUE (FT-LBS) | .2504819E+00 |
| HTDMCD- | DMCD WHEEL MOMENTUM REQUIR. | .5078648E+02 |

Figure 9. Calculated output values.

| 1. Report No. | 2. Government Accession No. | 3. Recipient's Catalog No. |
|---|---|---|
| NASA TM 81798 | | |

| 4. Title and Subtitle | 5. Report Date |
|---|---|
| AN INTERACTIVE COMPUTER PROGRAM FOR SIZING SPACECRAFT MOMENTUM STORAGE DEVICES | March 1980 |
| | 6. Performing Organization Code |
| 7. Author(s) | 8. Performing Organization Report No. |
| Floyd J. Wilcox, Jr.* | |
| | 10. Work Unit No. |
| 9. Performing Organization Name and Address | 540-02-13 |
| NASA Langley Research Center<br>Hampton, VA 23665 | 11. Contract or Grant No. |
| | 13. Type of Report and Period Covered |
| 12. Sponsoring Agency Name and Address | Technical Memorandum |
| National Aeronautics and Space Administration<br>Washington, DC 20546 | 14. Sponsoring Agency Code |

15. Supplementary Notes

*Cooperative education student at Georgia Institute of Technology

16. Abstract

An interactive computer program has been developed which computes the sizing requirements for nongimbled reaction wheels, control moment gyros (CMG), and dual momentum control devices (DMCD) used in Earth-orbiting spacecraft. The program accepts as inputs the spacecraft's environmental disturbance torques, rotational inertias, maneuver rates, and orbital data. From these inputs, wheel weights are calculated for a range of radii and rotational speeds. The shape of the momentum wheel may be chosen to be either a hoop, solid cylinder, or annular cylinder. The program provides graphic output illustrating the trade-off potential between the weight, radius, and wheel speed. A number of the intermediate calculations such as the X-, Y-, and Z-axis total momentum, the momentum absorption requirements for reaction wheels, CMG's, DMCD's, and basic orbit analysis information are also provided as program output.

| 17. Key Words (Suggested by Author(s)) | 18. Distribution Statement |
|---|---|
| Momentum storage<br>Spacecraft design | Unclassified - Unlimited<br><br>Subject Category 61,18 |

| 19. Security Classif. (of this report) | 20. Security Classif. (of this page) | 21. No. of Pages | 22. Price* |
|---|---|---|---|
| Unclassified | Unclassified | 39 | $4.50 |

N-305

*For sale by the National Technical Information Service, Springfield, Virginia 22161